

In-class exam

(solutions)

This exam consists of five parts that can be solved in any order.

1 First-order data types

For the first two questions, consider the following data type of bit strings:

data *Bits* = *BNil* | *B0 Bits* | *B1 Bits*

We interpret bit strings as representing natural numbers in binary like so:

$toInt :: Bits \rightarrow Int$
 $toInt\ BNil = 0$
 $toInt\ (B0\ x) = 2 * toInt\ x$
 $toInt\ (B1\ x) = 2 * toInt\ x + 1$

Question 1.1. Write a function $inc :: Bits \rightarrow Bits$ that increments the bit string representing a number. It should satisfy the property that $toInt\ (inc\ bs) = 1 + toInt\ bs$.

Solution :

$inc :: Bits \rightarrow Bits$
 $inc\ BNil = B1\ BNil$
 $inc\ (B0\ x) = B1\ x$
 $inc\ (B1\ x) = B0\ (inc\ x)$

We can verify by induction that $toInt\ (inc\ bs) = 1 + toInt\ bs$. Indeed the cases $bs = BNil$ and $bs = B0\ x$ are immediate, while for the case $bs = B1\ x$, we get

$$\begin{aligned}
 toInt\ (inc\ (B1\ x)) &= toInt\ (B0\ (inc\ x)) \\
 &= 2 * toInt\ (inc\ x) \\
 &= 2 * (1 + toInt\ x) \\
 &= 2 + 2 * toInt\ x \\
 &= 1 + toInt\ (B1\ x)
 \end{aligned}$$

by unrolling the definitions of inc and $toInt$ and applying the IH and a bit of algebra. (Although the question did not ask us to prove this property of inc , proving it helps us ensure that our function is correct!) \square

Question 1.2. This representation contains redundancies, since many different values of type *Bits* can represent the value 0. Write a function $nf :: Bits \rightarrow Bits$ that eliminates such redundancies by converting a bit string to *normal form*. It should satisfy the property that $toInt\ (nf\ x) = toInt\ x$, and that $toInt\ (nf\ x) = toInt\ (nf\ y)$ iff $nf\ x = nf\ y$.

Solution : Redundancies arise because $BNil$, $B0\ BNil$, $B0\ (B0\ BNil)$, etc., all represent the number 0. To bring the bit string to normal form, we can recursively inspect and reconstruct it, making sure never to introduce a value of the form $B0\ BNil$.

```

nf :: Bits → Bits
nf BNil = BNil
nf (B0 n) = case nf n of
  BNil → BNil
  x → B0 x
nf (B1 n) = B1 (nf n)

```

Again we can easily verify by induction that $\text{toInt} (\text{nf } x) = \text{toInt } x$, and that $\text{toInt} (\text{nf } x) = 0$ just in case $\text{nf } x = \text{BNil}$, from which we can derive that $\text{toInt} \circ \text{nf}$ is injective. \square


For the next question, consider the following data types of binary trees and of general (arbitrary branching) trees with labelled nodes:

```

data BinTree a = L | B a (BinTree a) (BinTree a)
data Tree a = Node a [Tree a]

```

The “left-child, right-sibling” encoding defines a correspondence between binary trees and general trees, by interpreting the left child of a binary node as a child and the right child as a sibling in the corresponding general tree. Formally, this interprets a binary tree as a *forest* of general trees, but one can turn a forest into a tree by adding a distinguished root label. We thus have a 1-1 correspondence between pairs (label, binary tree) and general trees:



Question 1.3. Write a pair of functions

```

toTree :: (a, BinTree a) → Tree a
fromTree :: Tree a → (a, BinTree a)

```

implementing the left-child, right-sibling correspondence, and prove that they define a type isomorphism $(a, \text{BinTree } a) \cong \text{Tree } a$.

Solution : We find it convenient to write toTree and fromTree in mutual recursion with a pair of functions

```

toForest :: BinTree a → [Tree a]
fromForest :: [Tree a] → BinTree a

```

converting between binary trees and forests:

```

toTree (x, t) = Node x (toForest t)
toForest L = []
toForest (B x t u) = toTree (x, t) : toForest u
fromTree (Node x ts) = (x, fromForest ts)
fromForest [] = L
fromForest (t : ts) = let (x, t') = fromTree t in B x t' (fromForest ts)

```

Likewise, to prove that *toTree* and *fromTree* define an isomorphism $(a, \text{BinTree } a) \cong \text{Tree } a$, we simultaneously prove by structural induction that *toForest* and *fromForest* define an isomorphism $\text{BinTree } a \cong [\text{Tree } a]$. The proof is a long but mechanical calculation:

$$\begin{aligned}
\text{fromTree } (\text{toTree } (x, t)) &= \text{fromTree } (\text{Node } x \text{ (toForest } t)) \\
&= (x, \text{fromTree } (\text{toForest } t)) \\
&= (x, t) \\
\text{fromForest } (\text{toForest } L) &= \text{fromForest } [] \\
&= L \\
\text{fromForest } (\text{toForest } (B \ x \ t \ u)) &= \text{fromForest } (\text{toTree } (x, t) : \text{toForest } u) \\
&= \text{let } (x', t') = \text{fromTree } (\text{toTree } (x, t)) \text{ in } B \ x' \ t' \ (\text{fromForest } (\text{toForest } u)) \\
&= \text{let } (x', t') = (x, t) \text{ in } B \ x' \ t' \ u \\
&= B \ x \ t \ u \\
\text{toTree } (\text{fromTree } (\text{Node } x \ ts)) &= \text{toTree } (x, \text{fromForest } ts) \\
&= \text{Node } x \ (\text{toForest } (\text{fromForest } ts)) \\
&= \text{Node } x \ ts \\
\text{toForest } (\text{fromForest } []) &= \text{toForest } L \\
&= [] \\
\text{toForest } (\text{fromForest } (t : ts)) &= \text{toForest } (\text{let } (x, t') = \text{fromTree } t \text{ in } B \ x \ t' \ (\text{fromForest } ts)) \\
&= \text{let } (x, t') = \text{fromTree } t \text{ in } \text{toForest } (B \ x \ t' \ (\text{fromForest } ts)) \\
&= \text{let } (x, t') = \text{fromTree } t \text{ in } \text{toTree } (x, t') : \text{toForest } (\text{fromForest } ts) \\
&= \text{toTree } (\text{fromTree } t) : \text{toForest } (\text{fromForest } ts) \\
&= t : ts
\end{aligned}$$

□

2 Higher-order functions

Question 2.1. Implement the standard library function $\text{maybe} :: b \rightarrow (a \rightarrow b) \rightarrow \text{Maybe } a \rightarrow b$, which internalizes the principle of case-analysis on a value of *Maybe* type.

Solution :

$$\begin{aligned}
\text{maybe } x \ f \ \text{Nothing} &= x \\
\text{maybe } x \ f \ (\text{Just } y) &= f \ y
\end{aligned}$$

□

Question 2.2. The standard library function $\text{lookup} :: \text{Eq } a \Rightarrow a \rightarrow [(a, b)] \rightarrow \text{Maybe } b$ finds the first value matching a given key in a list of (key, value) pairs. Express *lookup* in terms of the function $\text{find} :: (c \rightarrow \text{Bool}) \rightarrow [c] \rightarrow \text{Maybe } c$, which finds the first value satisfying a predicate in a list, and *maybe* from the previous question.

Solution :

$$\text{lookup } x = \text{maybe } \text{Nothing} \ (\text{Just} \circ \text{snd}) \circ \text{find } ((\equiv x) \circ \text{fst})$$

□

Question 2.3. Express *find* in terms of *foldr*.

Solution :

$$\text{find } p = \text{foldr } (\backslash x \text{ } mx \rightarrow \text{if } p \text{ } x \text{ then Just } x \text{ else } mx) \text{ Nothing}$$

□

Question 2.4. Prove the *fusion law* for *foldr*, which states that if h is a function such that $h \ e = e'$ and $h \ (f \ x \ y) = f' \ x \ (h \ y)$, then $h \circ \text{foldr } f \ e = \text{foldr } f' \ e'$.

Solution : Recall the definition of *foldr*:

$$\begin{aligned} \text{foldr } f \ e \ [] &= e \\ \text{foldr } f \ e \ (x : xs) &= f \ x \ (\text{foldr } f \ e \ xs) \end{aligned}$$

We prove that $h \ (\text{foldr } f \ e \ xs) = \text{foldr } f' \ e' \ xs$ for all xs , by induction on xs :

- (Case $xs = []$): then $h \ (\text{foldr } f \ e \ []) = h \ e = e' = \text{foldr } f' \ e' \ []$.
- (Case $xs = x:ys$): then $h \ (\text{foldr } f \ e \ (x:ys)) = h \ (f \ x \ (\text{foldr } f \ e \ ys)) = f' \ x \ (h \ (\text{foldr } f \ e \ ys)) = f' \ x \ (\text{foldr } f' \ e' \ ys) = \text{foldr } f' \ e' \ (x : ys)$.

□

Question 2.5. Using your answers to the previous questions, *derive* an expression for *lookup* using only *foldr*. (For full credit, you must justify the expression using the fusion law.)

Solution : By Q2.2 and Q2.3, we have that $\text{lookup } x = h \circ \text{foldr } f \ e$ where

$$\begin{aligned} f &= \backslash kv \text{ } mx \rightarrow \text{if } \text{fst } kv \equiv x \text{ then Just } kv \text{ else } mx \\ e &= \text{Nothing} \\ h &= \text{maybe Nothing } (\text{Just} \circ \text{snd}) \end{aligned}$$

We can calculate that

$$\begin{aligned} h \ (f \ kv \ mx) &= \text{if } \text{fst } kv \equiv x \text{ then Just } (\text{snd } kv) \text{ else } h \ mx \\ h \ e &= \text{Nothing} \end{aligned}$$

and hence we can conclude that

$$\text{lookup } x = \text{foldr } (\backslash kv \text{ } mx \rightarrow \text{if } \text{fst } kv \equiv x \text{ then Just } (\text{snd } kv) \text{ else } mx) \text{ Nothing}$$

by the fusion law.

□

3 λ -calculus and propositions-as-types

Question 3.1. For each of the Haskell functions below, state whether the function is typable, and if so give its most general type.

$$\begin{aligned} f1 &= \backslash x \rightarrow [x, x \circ x, x \circ x \circ x] \\ f2 &= \backslash x \ y \rightarrow (y, x + y) \\ f3 &= \backslash x \ y \rightarrow x + y \ x + y \end{aligned}$$

Solution :

$$f1 :: (a \rightarrow a) \rightarrow [a \rightarrow a]$$

We infer that the argument x of $f1$ must have type $a \rightarrow a$ since it is composed with itself, and we can verify that $[x, x \circ x, x \circ x \circ x]$ is a valid list of type $[a \rightarrow a]$.

$$f2 :: \text{Num } a \Rightarrow a \rightarrow a \rightarrow (a, a)$$

The arguments of $f2$ are added together so both must be numbers, and the function returns a pair of numbers.

$$f3 \text{ untypable}$$

The argument y of $f3$ must be both a number and a function from numbers to numbers, which is impossible. \square

Peirce's law is a propositional tautology stating that $((P \supset Q) \supset P) \supset P$. Although true classically, Peirce's law is not constructively valid, and there is no simply-typed lambda term whose principal type is $((p \rightarrow q) \rightarrow p) \rightarrow p$.

Question 3.2. Assume that you are given a polymorphic function $peirce :: ((p \rightarrow q) \rightarrow p) \rightarrow p$ in Haskell. Show how you could use it to define a polymorphic expression $lem :: \text{Either } p (p \rightarrow \text{Void})$ realizing the law of excluded middle. (Here Void is the data type with no constructors.)

Solution : Let's begin by describing a constructive proof in natural language. Assuming Peirce's law, we have in particular (by letting P be $P \vee \neg P$ and Q be \perp) that

$$(((P \vee \neg P) \supset \perp) \supset (P \vee \neg P)) \supset (P \vee \neg P) \tag{1}$$

and so to derive the law of excluded middle $P \vee \neg P$ it suffices to prove that

$$((P \vee \neg P) \supset \perp) \supset (P \vee \neg P) \tag{2}$$

and apply modus ponens to (1) and (2). To show (2), we assume

$$(P \vee \neg P) \supset \perp \tag{3}$$

and derive $P \vee \neg P$, for which it suffices to show $\neg P$. So assume by way of contradiction that

$$P \tag{4}$$

is true. Then $P \vee \neg P$ is true. But that contradicts the hypothesis (3). QED.

The argument above translates directly to the following Haskell code, where hypothesis (3) corresponds to the variable f , and (4) to the variable x :

$$lem = peirce (\backslash f \rightarrow \text{Right } (\backslash x \rightarrow f (\text{Left } x)))$$

\square

4 Side-effects and monads

For the next few questions, consider the type

data *Logged* *a* = *Logged* *a* [*String*]

whose values consist of values of type *a* paired with a list of strings. We can think of values of type *Logged a* as computations producing a sequence of string outputs on the way to a value.

Question 4.1. Show *Logged* is a functor, by defining $fmap :: (a \rightarrow b) \rightarrow Logged\ a \rightarrow Logged\ b$ and verifying that $fmap\ id = id$ and $fmap\ (f \circ g) = fmap\ f \circ fmap\ g$ for all *f* and *g*.

Solution :

$$fmap\ f\ (L\ x\ l) = L\ (f\ x)\ l$$

The functor laws are verified immediately. □

Question 4.2. Turn *Logged* into a monad by defining

$return :: a \rightarrow Logged\ a$
 $(\gg=) :: Logged\ a \rightarrow (a \rightarrow Logged\ b) \rightarrow Logged\ b$

so that *return* creates a pure computation with an empty log, while $(\gg=)$ runs a logged computation and feeds the result to a log-producing continuation, concatenating the two logs. Then, prove that these definitions satisfy the three monad laws

$$(return\ x \gg= f) = f\ x \quad (lx \gg= return) = lx \quad (lx \gg= f) \gg= g = lx \gg= (\lambda x \rightarrow f\ x \gg= g)$$

for all $x :: a$, $lx :: Logged\ a$, $f :: a \rightarrow Logged\ b$, and $g :: b \rightarrow Logged\ c$.

Solution :

$return\ x = Logged\ x\ []$
 $Logged\ x\ l1 \gg= f = \mathbf{let}\ Logged\ y\ l2 = f\ x\ \mathbf{in}\ Logged\ y\ (l1 ++ l2)$

We verify the monad laws:

$$\begin{aligned} return\ x \gg= f &= L\ x\ [] \gg= f \\ &= \mathbf{let}\ Logged\ y\ l = f\ x\ \mathbf{in}\ Logged\ y\ ([] ++ l) \\ &= f\ x \\ (Logged\ x\ l1 \gg= return) &= \mathbf{let}\ Logged\ y\ l2 = Logged\ x\ []\ \mathbf{in}\ Logged\ y\ (l1 ++ l2) \\ &= Logged\ x\ (l1 ++ []) \\ &= Logged\ x\ l1 \\ (Logged\ x\ l1 \gg= f) \gg= g &= (\mathbf{let}\ Logged\ y\ l2 = f\ x\ \mathbf{in}\ Logged\ y\ (l1 ++ l2)) \gg= g \\ &= \mathbf{let}\ Logged\ y\ l2 = f\ x\ \mathbf{in}\ (Logged\ y\ (l1 ++ l2) \gg= g) \\ &= \mathbf{let}\ Logged\ y\ l2 = f\ x\ \mathbf{in}\ \mathbf{let}\ Logged\ z\ l3 = g\ y\ \mathbf{in}\ Logged\ z\ ((l1 ++ l2) ++ l3) \\ &= \mathbf{let}\ Logged\ y\ l2 = f\ x\ \mathbf{in}\ \mathbf{let}\ Logged\ z\ l3 = g\ y\ \mathbf{in}\ Logged\ z\ (l1 ++ (l2 ++ l3)) \\ &= Logged\ x\ l1 \gg= (\lambda x \rightarrow f\ x \gg= g) \end{aligned}$$

□

Question 4.3. Define a function $\text{log} :: \text{String} \rightarrow \text{Logged } ()$ representing a computation that writes a single string to the log and returns a trivial value.

Solution :

```
log s = L () [s]
```

□

Question 4.4. Consider the following data type of arithmetic expressions:

```
data Exp = Con Double | Add Exp Exp | Mul Exp Exp
```

Define a function $\text{evalLogged} :: \text{Exp} \rightarrow \text{Logged Double}$ that evaluates an expression to a number together with a log recording the order in which the subexpressions are evaluated.

Solution :

```
evalLogged e = do
  log ("evaluating: " ++ show e)
  case e of
    Con x → return x
    Add e1 e2 → do
      x1 ← evalLogged e1
      x2 ← evalLogged e2
      return (x1 + x2)
    Mul e1 e2 → do
      x1 ← evalLogged e1
      x2 ← evalLogged e2
      return (x1 * x2)
```

□

5 Laziness and infinite objects

The goal of this last section is to write some *seemingly impossible functional programs*.¹ The *Cantor space* 2^ω is a topological space whose elements are infinite sequences of binary digits. To formalize the Cantor space in Haskell, let's begin by recalling the (co)data type *Stream a* of infinite sequences of *as*:

```
data Stream a = Stream { hd :: a, tl :: Stream a }
```

We then define the Cantor space as the type of streams of bits:

```
data Bit = Zero | One deriving (Eq)
type Cantor = Stream Bit
```

Question 5.1. Define an operation

```
(#) :: Bit → Cantor → Cantor
```

that prepends a bit to the beginning of an infinite sequence of bits.

¹Adapted from Martín Escardó's guest article of the same name on the *Mathematics and Computation* blog (Escardó, September 2007).

Solution :

$$b \# bs = \text{Stream } \{hd = b, tl = bs\}$$

□

To gain some more intuition, let's show that values of type *Cantor* may be interpreted as functions from natural numbers to bits, and vice versa. (For convenience, we'll write *Nat* as a type synonym for *Int*, but with the understanding that its values are restricted to non-negative integers.)

Question 5.2. Implement a pair of coercions $\text{fromCantor} :: \text{Cantor} \rightarrow (\text{Nat} \rightarrow \text{Bit})$ and $\text{toCantor} :: (\text{Nat} \rightarrow \text{Bit}) \rightarrow \text{Cantor}$. Your functions should realize a type isomorphism $\text{Cantor} \cong \text{Nat} \rightarrow \text{Bit}$, although you do not need to prove this fact.

Solution :

$$\begin{aligned} &\text{fromCantor } s \ n \\ &\quad | \ n \equiv 0 = \text{hd } s \\ &\quad | \ n > 0 = \text{fromCantor } (tl \ s) \ (n - 1) \\ &\text{toCantor } f = \text{Stream } \{hd = f \ 0, tl = \text{toCantor } (\backslash n \rightarrow f \ (n + 1))\} \end{aligned}$$

□

Now, our goal will be to write a higher-order function

$$\text{existsC} :: (\text{Cantor} \rightarrow \text{Bool}) \rightarrow \text{Bool}$$

which takes a total predicate over the Cantor space, and decides if there is some infinite sequence of bits making the predicate true. By *total* predicate, we mean a function $\text{Cantor} \rightarrow \text{Bool}$ that terminates for all input values. For example,

$$\text{good } s = \text{fromCantor } s \ 3 \neq \text{fromCantor } s \ 4 \ \&\& \ \text{fromCantor } s \ 7 \equiv \text{fromCantor } s \ 15$$

is a total predicate (testing that bits 3 and 4 are distinct, and bits 7 and 15 are equal), but

$$\text{bad } s = \text{hd } s \equiv \text{One} \ \&\& \ \text{bad } (tl \ s)$$

is not total, since it will not terminate on an infinite sequence of *Ones*.

Question 5.3. Write *existsC* in mutual recursion with a function

$$\text{findC} :: (\text{Cantor} \rightarrow \text{Bool}) \rightarrow \text{Cantor}$$

so that they satisfy the following specifications:

- For any total predicate p , $\text{existsC } p$ should terminate and return *True* just in case there exists a sequence $bs :: \text{Cantor}$ such that $p \ bs = \text{True}$, or else return *False*.
- For any total predicate p , $\text{findC } p$ should terminate and return a sequence $bs :: \text{Cantor}$ such that $p \ bs = \text{True}$ if there exists such a sequence, or else return an arbitrary sequence.

Hint: Put your faith in the specifications to define these functions by mutual recursion. Keep in mind that findC needs to return an infinite sequence of bits, so in particular it needs to determine the initial bit.

Solution :

```
existsC p = p (findC p)
findC p = if existsC (\bs → p (Zero # bs)) then Zero # findC (\bs → p (Zero # bs))
         else One # findC (\bs → p (One # bs))
```

(For a deeper discussion of these definitions, see Escardó's article mentioned above.) □

Question 5.4. Define an operation

$$\text{equalC} :: Eq\ a \Rightarrow (Cantor \rightarrow a) \rightarrow (Cantor \rightarrow a) \rightarrow Bool$$

that takes two total functions over the Cantor space into some type admitting decidable equality, and decides whether the functions are equal on all inputs. You can assume as given *existsC* and *findC* satisfying the specifications from the previous question.

Solution :

$$\text{equalC}\ f\ g = \text{not}\ (\text{existsC}\ (\lambda c \rightarrow f\ c \neq g\ c))$$

To check that two functions are equal, we test that they do not disagree on any input sequence. (Remarkably, *equalC* *f* *g* is well-defined and always terminates, even though it looks like there are infinitely many, infinitely long sequences to test!) □