

# **[CSE301 / Lecture 4]**

## **Side-effects and monads**

Noam Zeilberger

Ecole Polytechnique


24 September 2025

## What are side-effects?


Everything that a function does besides computing a functional relation from inputs to outputs.

In other words, the difference between “functions in math” and “functions in Python”.

## What are side-effects?



*versus*



**“pure” function = no side-effects**

```
def sqr(x):  
    y = x ** 2  
    return y
```

---

```
>>> sqr(3)  
9  
>>> sqr(4)  
16
```

## Printing debugging information

```
def sqr_debug(x):  
    y = x ** 2  
    print("Squaring {} gives {}!!".format(x, y))  
    return y
```

---

```
>>> sqr_debug(3)  
Squaring 3 gives 9!!  
9  
>>> sqr_debug(4)  
Squaring 4 gives 16!!  
16
```

## Getting input from the user, and raising exceptions

```
def exp_by_input(x):  
    k = int(input("Enter exponent: "))  
    y = x ** k  
    return y
```

---

```
>>> exp_by_input(3)
```

```
Enter exponent: 2
```

```
9
```

```
>>> exp_by_input(3)
```

```
Enter exponent: two
```

```
Traceback (most recent call last):
```

```
  File "<stdin>", line 1, in <module>
```

```
    File "<string>", line 7, in exp_by_input
```

```
ValueError: invalid literal for int() with base 10: 'two'
```

## Querying a random number generator

```
def exp_by_random(x):  
    k = random.randrange(0,10)  
    y = x ** k  
    return y
```

---

```
>>> exp_by_random(3)  
2187  
>>> exp_by_random(3)  
243
```

## Reading and writing global variables

```
k = 0
def exp_by_counter(x):
    global k
    y = x ** k
    k = k + 1
    return y
```

---

```
>>> exp_by_counter(3)
1
>>> exp_by_counter(3)
3
>>> exp_by_counter(3)
9
```



## Non-standard control flow (e.g., nondeterminism via generators)

```
def exp_by_nondet(x):  
    for k in range(0,10):  
        y = x ** k  
        yield y
```

---

```
>>> exp_by_nondet(3)  
<generator object exp_by_nondet at 0x7ff77d38c830>  
>>> list(exp_by_nondet(3))  
[1, 3, 9, 27, 81, 243, 729, 2187, 6561, 19683]
```

## Side-effects in Haskell

Despite claims, Haskell is not really a pure language...


1. Functions may not terminate
2. Functions may raise exceptions
3. Run-time performance can vary wildly due to laziness

Nevertheless, these effects (at least 1 & 2) are relatively “benign”.


In Haskell, most “serious” effects (like getting input from the user, or reading and writing global variables) are confined to *monads*.

## The idea, very roughly

Replace



by



where  $M_{world}$  captures all possible interactions with the world.

We'll make this more precise, but first let's talk a bit about the principles of *referential transparency* and *compositionality*...

## The principle of referential transparency

Informal principle that we can replace an expression by the value it computes without changing the behavior of a program, e.g.:

```
>>> sqr(3)
```

```
9
```

```
>>> 9 == 9
```

```
True
```

```
>>> sqr(3) == 9
```

```
True
```

```
>>> sqr(3) == sqr(3)
```

```
True
```

## The principle of referential transparency

The presence of side-effects can break referential transparency!

```
>>> exp_by_counter(3)
```

```
9
```

```
>>> exp_by_counter(3) == 9
```

```
False
```

```
>>> exp_by_counter(3) == exp_by_counter(3)
```

```
False
```

## Compositional semantics

More generally, a *semantics* for a programming language is a way of assigning meanings to program expressions. A desired property of a semantics is that it is *compositional*, in the sense that the meaning of an expression is built from the meanings of its subexpressions.

The presence of side-effects presents a challenge to defining compositional semantics for programming languages!<sup>1</sup>

---

<sup>1</sup>And this is also true for natural languages! See, for example, Chung-chieh Shan's PhD thesis, *Lingustic side effects* (2005).

## Toy example:<sup>2</sup> arithmetic expressions

Consider a little language of arithmetic expressions, with constants, subtraction, and division:

$$e ::= c \mid e1 - e2 \mid e1 / e2$$

Each expression  $e$  denotes a number  $\llbracket e \rrbracket \in \mathbb{R}$ , defined inductively:

$$\begin{aligned}\llbracket c \rrbracket &= c \\ \llbracket e1 - e2 \rrbracket &= \llbracket e1 \rrbracket - \llbracket e2 \rrbracket \\ \llbracket e1 / e2 \rrbracket &= \llbracket e1 \rrbracket / \llbracket e2 \rrbracket\end{aligned}$$

Division by zero is undefined, so  $\llbracket e \rrbracket$  is sometimes undefined.

---

<sup>2</sup>Inspired in part by Philip Wadler, “Monads for functional programming”, Proceedings of the Båstad Spring School, May 1995.

## Toy example: arithmetic expressions

Translated to Haskell:

```
data Expr = Con Double | Sub Expr Expr | Div Expr Expr
```

```
eval :: Expr → Double
```

```
eval (Con c) = c
```

```
eval (Sub e1 e2) = eval e1 - eval e2
```

```
eval (Div e1 e2) = eval e1 / eval e2
```



## Toy example: arithmetic expressions

Example expressions:

$$e1 = \text{Sub} (\text{Div} (\text{Con } 2) (\text{Con } 4)) (\text{Con } 3)$$

$$e2 = \text{Sub} (\text{Con } 1) (\text{Div} (\text{Con } 2) (\text{Con } 2))$$

$$e3 = \text{Div} (\text{Con } 1) (\text{Sub} (\text{Con } 2) (\text{Con } 2))$$

And their semantics:

$$\text{eval } e1 = -2.5$$

$$\text{eval } e2 = 0$$

$$\text{eval } e3 \text{ undefined}$$

## Variation #1: error-handling

Modify the semantics to handle division-by-zero.

In a language with exceptions, we could simply raise an exception. Haskell has them, but let's pretend it doesn't and stay "pure"...

Idea:  $e$  no longer denotes a number, but a "number or error".

That is,  $\llbracket e \rrbracket \in \mathbb{R} \uplus \{error\}$

In Haskell, we can return a Maybe type...

## Variation #1: error-handling

```
eval1 :: Expr → Maybe Double
eval1 (Con c)    = Just c
eval1 (Sub e1 e2) =
  case (eval1 e1, eval1 e2) of
    (Just x1, Just x2) → Just (x1 - x2)
    _                 → Nothing
eval1 (Div e1 e2) =
  case (eval1 e1, eval1 e2) of
    (Just x1, Just x2)
      | x2 /= 0 → Just (x1 / x2)
      | otherwise → Nothing
    _          → Nothing
```

## Variation #1: error-handling

The example expressions:

$$e1 = \text{Sub } (\text{Div } (\text{Con } 2) (\text{Con } 4)) (\text{Con } 3)$$
$$e2 = \text{Sub } (\text{Con } 1) (\text{Div } (\text{Con } 2) (\text{Con } 2))$$
$$e3 = \text{Div } (\text{Con } 1) (\text{Sub } (\text{Con } 2) (\text{Con } 2))$$

In the new semantics:

$$\text{eval1 } e1 = \text{Just } (-2.5)$$
$$\text{eval1 } e2 = \text{Just } 0.0$$
$$\text{eval1 } e3 = \text{Nothing}$$

## Variation #2: global state

Modify the semantics of expressions so that every third constant is interpreted as 0. (Yeah this is a bit weird, but so is most of life.)

The meaning of a subexpression now depends on its position. E.g.,  $\llbracket 3 - 2 \rrbracket = \llbracket 1 \rrbracket$ , but  $\llbracket 6 / (3 - 2) \rrbracket = \llbracket 6 / (3 - 0) \rrbracket \neq \llbracket 6 / 1 \rrbracket$ .

Can we define a compositional semantics?...

## Variation #2: global state

...Yes, in **state-passing style**!

Idea: every subexpression  $e$  denotes a function  $\llbracket e \rrbracket \in \mathbb{N} \rightarrow \mathbb{R} \times \mathbb{N}$  taking a count of the previously seen constants, and returning a number together with an updated count.

For the top-level expression, initialize count to 0.

## Variation #2: global state

```
eval2 :: Expr → Int → (Double, Int)
eval2 (Con c) n =
  (if mod n 3 == 2 then 0 else c, n + 1)
eval2 (Sub e1 e2) n =
  let (x1, o) = eval2 e1 n in
  let (x2, p) = eval2 e2 o in
  (x1 - x2, p)
eval2 (Div e1 e2) n =
  let (x1, o) = eval2 e1 n in
  let (x2, p) = eval2 e2 o in
  (x1 / x2, p)
```

```
eval2Top :: Expr → Double
eval2Top e = fst (eval2 e 0)
```

## Variation #2: global state

The example expressions:

$$e1 = \text{Sub} (\text{Div} (\text{Con } 2) (\text{Con } 4)) (\text{Con } 3)$$

$$e2 = \text{Sub} (\text{Con } 1) (\text{Div} (\text{Con } 2) (\text{Con } 2))$$

$$e3 = \text{Div} (\text{Con } 1) (\text{Sub} (\text{Con } 2) (\text{Con } 2))$$

In the new semantics:

$$\text{eval2Top } e1 = 0.5$$

$$\text{eval2Top } e2 \text{ undefined}$$

$$\text{eval2Top } e3 = 0.5$$



### Variation #3: combining error-handling and state

```
eval3 :: Expr → Int → Maybe (Double, Int)
eval3 (Con c) n =
  Just (if mod n 3 == 2 then 0 else c, n + 1)
eval3 (Sub e1 e2) n =
  case eval3 e1 n of
    Nothing → Nothing
    Just (x1, o) → case eval3 e2 o of
      Nothing → Nothing
      Just (x2, p) → Just (x1 - x2, p)
eval3 (Div e1 e2) n =
  case eval3 e1 n of
    Nothing → Nothing
    Just (x1, o) → case eval3 e2 o of
      Nothing → Nothing
      Just (x2, p)
        | x2 /= 0 → Just (x1 / x2, p)
        | otherwise → Nothing
```

## Variation #3: combining error-handling and state

$eval3Top :: Expr \rightarrow Maybe Double$   
 $eval3Top\ e = \mathbf{case}\ eval3\ e\ \mathbf{of}$   
     $Nothing \rightarrow Nothing$   
     $Just\ (x, \_)\rightarrow Just\ x$

In this last semantics:

$eval3Top\ e1 = Just\ 0.5$   
 $eval3Top\ e2 = Nothing$   
 $eval3Top\ e3 = Just\ 0.5$

## Compare with the OCaml version...

```
type expr = Con of float
          | Sub of expr * expr | Div of expr * expr
let cnt = ref 0
let rec eval3 (e : expr) : float =
  match e with
  | Con c -> let n = !cnt in
              (cnt := n+1; if n mod 3 == 2 then 0.0 else c)
  | Sub (e1,e2) -> let x1 = eval3 e1 in
                   let x2 = eval3 e2 in
                   x1 -. x2
  | Div (e1,e2) -> let x1 = eval3 e1 in
                   let x2 = eval3 e2 in
                   if x2 <> 0.0 then x1 /. x2
                   else raise Division_by_zero
let rec eval3Top e = (cnt := 0; eval3 e)
```

## Haskell version #3, rewritten using a monad and do notation

```
eval3' :: Expr → StateT Int Maybe Double
eval3' (Con c) = do
  n ← get
  put (n + 1)
  return (if mod n 3 == 2 then 0 else c)
eval3' (Sub e1 e2) = do
  x1 ← eval3' e1
  x2 ← eval3' e2
  return (x1 - x2)
eval3' (Div e1 e2) = do
  x1 ← eval3' e1
  x2 ← eval3' e2
  if x2 /= 0 then return (x1 / x2) else lift Nothing
eval3' Top e = runStateT (eval3' e) 0 >>= return ∘ fst
```

# What is a monad?

A concept originating in category theory.

Proposed by Eugenio Moggi as a unifying mathematical model of different notions of computation.

Adapted by Phil Wadler as a way of integrating side-effects with pure functional programming, in particular in Haskell.

# What is a category?

A *category* is a gadget consisting of the following:

- a set of objects;
- for every pair of objects, a set of arrows between them;
- for every object  $a$ , a distinguished *identity* arrow  $id_a : a \rightarrow a$ ;
- for every pair of arrows  $f : a \rightarrow b$  and  $g : b \rightarrow c$ , a *composite* arrow  $g \circ f : a \rightarrow c$ ;
- such that composition and identity are associative and unital in the appropriate sense.

Intuition: the arrows of a category can be seen as describing a collection of “allowable transformations” between objects.

# What is a category?

Examples:

- Set : sets as objects, arrows are *total functions*  $f : X \rightarrow Y$
- Rel : sets as objects, arrows are *arbitrary relations*  $r \subseteq X \times Y$
- Grp : groups as objects, arrows *homomorphisms*  $\phi : G \rightarrow H$
- Vec : vector spaces as objects, arrows *linear maps*  $\phi : V \rightarrow W$

## A category of pure Haskell functions

Informally, we can think of Haskell functions as forming the arrows of a category, whose objects are types.

$$Int \xrightarrow{(+1)} Int \quad Int \xrightarrow{(>0)} Bool \quad etc.$$

Composition of arrows is defined by function composition.

$$Int \xrightarrow{(+1)} Int \xrightarrow{(>0)} Bool$$

$$\quad \quad \quad \searrow \quad \quad \quad \nearrow$$

$$\quad \quad \quad (>0) \circ (+1)$$

The identity function  $\lambda x \rightarrow x$  serves as the identity arrow  $a \rightarrow a$ .



## Beyond the category of pure functions

But... we don't always want to program inside this category!

Monads give us a way of building new categories to program in.

A monad is a special kind of *functor*.

# What is a functor?

A *functor* is a way of mapping one category into another:

- It should map objects to objects and arrows to arrows.
- It should preserve identity and composition.

Examples:

- $|-| : \mathbf{Grp} \rightarrow \mathbf{Set}$  sending a group  $G$  to its underlying carrier  $|G|$ , and a homomorphism  $\phi : G \rightarrow H$  to the underlying function between carriers  $|\phi| : |G| \rightarrow |H|$
- $P : \mathbf{Set} \rightarrow \mathbf{Set}$  sending a set  $X$  to its powerset  $P(X)$ , and a function  $f : X \rightarrow Y$  to the function  $P(f) : P(X) \rightarrow P(Y)$  defined by  $P(f) = \lambda S. \{f(x) \mid x \in S\}$ .

# Functors in Haskell

The Functor type class:

```
class Functor f where  
    fmap :: (a → b) → f a → f b
```

Note here  $f$  is a type constructor  $f :: * \rightarrow *$ .

Any instance should satisfy the *functor laws*:

$$fmap\ id = id \qquad fmap\ (f \circ g) = fmap\ f \circ fmap\ g$$

## Example #1: the List functor

The list type constructor is a functor:

```
instance Functor [] where  
  -- fmap :: (a -> b) -> [a] -> [b]  
  fmap = map
```

(Exercise: prove the functor laws  $\text{map id } xs = xs$  and  $\text{map } (f \circ g) \text{ } xs = \text{map } f (\text{map } g \text{ } xs)$  by structural induction!)

## Example #2: the Maybe functor

The *Maybe* type constructor is a functor:

**instance** *Functor Maybe* **where**

  -- fmap :: (a -> b) -> Maybe a -> Maybe b

*fmap f Nothing = Nothing*

*fmap f (Just x) = Just (f x)*

## Rough definition of a monad, in category theory

A *monad* is a functor  $m$  from a category to itself, equipped with an arrow  $a \rightarrow m a$  for every object  $a$ , together with a way of transforming arrows  $a \rightarrow m b$  into arrows  $m a \rightarrow m b$ , subject to certain equations.

For example, the powerset functor  $P : \text{Set} \rightarrow \text{Set}$  is a monad: the functions  $X \rightarrow P(X)$  are defined by taking singletons, and any function  $X \rightarrow P(Y)$  extends to a function  $P(X) \rightarrow P(Y)$  by taking unions.

## Kleisli category

One motivation for the definition of monad is that it allows to build a new category with the same objects, but where an arrow  $a \rightarrow b$  in the new category corresponds to an arrow  $a \rightarrow m\,b$  in the old category. To compose an arrow  $f : a \rightarrow m\,b$  with an arrow  $g : b \rightarrow m\,c$ , we first transform the latter to  $g^* : m\,b \rightarrow m\,c$  and then use ordinary composition to get  $g^* \circ f : a \rightarrow m\,c$ .

For, the Kleisli category of the powerset monad  $P$  on  $\mathbf{Set}$  recovers the category of sets and relations  $\mathbf{Rel} \cong \mathbf{Set}_P$ .

## Monads in Haskell (before 2014)

The Monad type class:

**class** *Monad* *m* **where**

*return* :: *a* → *m a*

(*>>=*) :: *m a* → (*a* → *m b*) → *m b*

Subject to the *monad laws*:

*return* *x* *>>=* *f* = *f* *x*

*mx* *>>=* *return* = *mx*

(*mx* *>>=* *f*) *>>=* *g* = *mx* *>>=* (*\x* → (*f* *x* *>>=* *g*))

(Note that *flip* (*>>=*) :: (*a* → *m b*) → (*m a* → *m b*).)



## The List and Maybe monads

**instance** *Monad* [] **where**

-- return :: a -> [a]

*return* x = [x]

-- (»=) :: [a] -> (a -> [b]) -> [b]

*xs* »= *f* = *concatMap* *f* *xs*

**instance** *Monad* *Maybe* **where**

-- return :: a -> Maybe a

*return* x = *Just* x

-- (»=) :: Maybe a -> (a -> Maybe b) -> Maybe b

*Nothing* »= *f* = *Nothing*

*Just* x »= *f* = *f* x

# Monads as notions of computation

Via the Kleisli category constructions...

- List monad: category of nondeterministic functions  $a \rightarrow [b]$ .
- Maybe monad: category of partial functions  $a \rightarrow \textit{Maybe } b$ .

## Evaluator #1, re-expressed using the Maybe monad

$eval1' :: Expr \rightarrow Maybe Double$

$eval1' (Con\ c) = return\ c$

$eval1' (Sub\ e1\ e2) =$

$eval1'\ e1 \gg= \backslash x1 \rightarrow$

$eval1'\ e2 \gg= \backslash x2 \rightarrow$

$return\ (x1 - x2)$

$eval1' (Div\ e1\ e2) =$

$eval1'\ e1 \gg= \backslash x1 \rightarrow$

$eval1'\ e2 \gg= \backslash x2 \rightarrow$

**if**  $x2 \neq 0$  **then**  $return\ (x1 / x2)$  **else** *Nothing*

## The State monad

The type constructor *State s* is defined essentially as follows:

```
newtype State s a = State { runState :: s → (a, s) }
```

It is a monad:

```
instance Monad (State s) where  
  return x = State (\s → (x, s))  
  xt >>= f = State (\s0 →  
    let (x, s1) = runState xt s0 in  
    runState (f x) s1)
```

## The State monad

Also, it supports “get” and “set” operations:

$$get :: State\ s\ s$$
$$get = State\ (\ s \rightarrow (s, s))$$
$$put :: s \rightarrow State\ s\ ()$$
$$put\ s' = State\ (\ s \rightarrow ((), s'))$$

## Evaluator #2, redefined using the State monad

```
eval2' :: Expr → State Int Double
eval2' (Con c) =
  get >>= \n →
  put (n + 1) >>= \_ →
  return (if mod n 3 == 2 then 0 else c)
eval2' (Sub e1 e2) =
  eval2' e1 >>= \x1 →
  eval2' e2 >>= \x2 →
  return (x1 - x2)
eval2' (Div e1 e2) =
  eval2' e1 >>= \x1 →
  eval2' e2 >>= \x2 →
  return (x1 / x2)
eval2Top' e = fst (runState (eval2' e) 0)
```

## Do notation

**do**  $x1 \leftarrow e1$   
     $x2 \leftarrow e2$   
    ...  
     $xn \leftarrow en$   
     $f\ x1\ x2\ \dots\ xn$

is syntactic sugar for

$e1 \gg= \backslash x1 \rightarrow$   
 $e2 \gg= \backslash x2 \rightarrow$   
...  
 $en \gg= \backslash xn \rightarrow$   
 $f\ x1\ x2\ \dots\ xn$

## Evaluator #2, equivalently expressed with do notation

$eval2' :: Expr \rightarrow State\ Int\ Double$

$eval2' (Con\ c) = \mathbf{do}$

$n \leftarrow get$

$put\ (n + 1)$

$return\ (\mathbf{if}\ mod\ n\ 3 == 2\ \mathbf{then}\ 0\ \mathbf{else}\ c)$

$eval2' (Sub\ e1\ e2) = \mathbf{do}$

$x1 \leftarrow eval2'\ e1$

$x2 \leftarrow eval2'\ e2$

$return\ (x1 - x2)$

$eval2' (Div\ e1\ e2) =$

$x1 \leftarrow eval2'\ e1$

$x2 \leftarrow eval2'\ e2$

$return\ (x1 / x2)$



## The IO monad

A built-in monad used to perform real system I/O.

Supports operations like

```
getLine :: IO String  
putStrLn :: String → IO ()
```

etc.

The use of a monad ensures proper sequentialization, as we can never “escape” the IO monad!<sup>3</sup>

---

<sup>3</sup>Technically, this is not true. There is a back door in the form of a function *unsafePerformIO* :: *IO a* → *a*, contained in *System.IO.Unsafe*. As the name suggests, this function should be used with care...

## Monads in Haskell (post 2014)

A bit more heavy since the “Functor-Applicative-Monad” hierarchy:

```
class Functor f where  
    fmap :: (a → b) → f a → f b  
  
class Functor f ⇒ Applicative f where  
    pure :: a → f a  
    (<*>) :: f (a → b) → f a → f b  
  
class Applicative m ⇒ Monad m where  
    return :: a → m a  
    (>=>) :: m a → (a → m b) → m b  
    return = pure
```

So to define an instance of `Monad`, you first need instances of `Functor` and `Applicative`.

## Monads in Haskell (post 2014)

But instances of Functor and Applicative can always be retrofitted from a Monad instance:

**instance** *Functor* *M* **where**

*fmap* *f* *xm* = *xm*  $\gg=$  *return*  $\circ$  *f*

**instance** *Applicative* *M* **where**

*pure* = *return*

*fm*  $\langle * \rangle$  *xm* = *fm*  $\gg=$   $\backslash f \rightarrow xm \gg= return \circ f$

## Combining monads

To define Evaluator #3, we implicitly used a *monad transformer*:

```
newtype StateT s m a =  
  StateT { runStateT :: s → m (a, s) }
```

Given a monad  $m$  representing some notion of computation (e.g., partiality or nondeterminism),  $\text{StateT } s \ m$  defines a new monad with  $s$  state wrapped around an  $m$ -computation.

But it is not always clear how to combine monads.

More generally, the question of how to organize and reason about programs with side-effects remains an important open problem!